
Тут я постараюсь написать руководство по скриптам для начинающего пользователя, который знать не знает, что такое программирование и вообще в первый раз имеет дело с написанием программ, так что извините, если некоторые вещи вам покажутся тривиальными, откровенно тупыми и даже вопиющей неправдой. Единственное, что понадобится для изучения – базовое знание OLC. Для полноценной работы со скриптами вам понадобится включенная опция wiznet script (выдает ошибки скриптов).
0. Что такое скрипты?

Что есть скрипты? Скрипты есть маленькая(или не очень) программа, которая заставляет мад как-то реагировать на определенные события. Скрипты в нашем мире являются альтернативой стандартным ромовским прогсам. У скриптов имеется множество достоинств (например, использование циклов, также из-за них не падает мир и т.п.) Прогсы из своих зон я советую исключать вообще.
0.1 Типы скриптов

Все скрипты делятся на три типа: мобскрипты, румскрипты, и обжскрипты. Соответственно, первые выполняются мобами (NPC), вторые – комнатами, третьи – объектами.

0.2 Как создать скрипт

Скрипт можно создать с помощью следующих команд:

mpedit create <vnum>
- создает мобпрогс;

rpedit create <vnum>
- создает румпрогс;

opedit create <vnum>
- создает обжпрогс.

Вместо <vnum> вам надо будет написать соответствующий внум прогса. Когда вы наберете эту команду, вы окажетесь внутри редактора скрипта. Чтобы компилятор воспринимал код, который вы напишете, как скрипты, вам необходимо набрать
version 1

Иначе компилятор будет считать скрипт стандартным ромовским прогсом. Вы также можете добавить скрипту ключевые слова, чтобы облегчить поиск нужного скрипта в будущем. Чтобы сделать это, воспользуйтесь командой keyword:

keyword <ключевые слова>
Настоятельная рекомендация: неработающие скрипты помечать ключевыми словами красного цвета.

Чтобы начать вводить код скрипта, наберите code. Вы окажетесь в редакторе кода.

Вы можете набрать .h внутри редактора кода, чтобы получить более подробную информацию о его работе.

NB Обратите внимание, что внутри редактора кода любой текст, кроме строчек, начинающихся с точки или знака «*», будет введен как код скрипта, так что отключайте триггера, когда находитесь внутри редактора! Строка, начинающаяся со знака «*» обозначает комментарий.
Чтобы выйти из редактора кода, введите символ «@».

Чтобы войти в редактор уже созданного скрипта, вам надо набрать
rpedit/mpedit/opedit <vnum>
Чтобы просмотреть список доступных скриптов в зоне, войдите в любой редатор в переделах конкретной зоны (например, для зоны с внумами 11400-11499, redit 11401) и наберите
rpedit/mpedit/opedit list
0.3 Как подключить скрипт к мобу/комнате/объекту

Чтобы подключить скрипт к мобу, комнате или объекту, вам надо войти в редактор моба, комнаты или объекта соответственно и набрать одну из следующих команд:

addmprog <vnum> <триггер> <аргумент триггера>;

addrprog <vnum> <триггер> <аргумент триггера>;

addoprog <vnum> <триггер> <аргумент триггера>.
NB Обратите внимание, что к мобу можно подключить только мобскрипт, к комнате – только румскрипт, а к объекту – только обжскрипт.

0.4 Что такое триггер и аргумент
Триггер – параметр, который указывает, при каком классе действий должен запуститься скрипт. Список триггеров, доступных для данного типа прогсов, вы можете узнать, зайдя в любой редактор (например, redit) и набрав
? mprog
? rprog
? oprog
соответственно. Каждый триггер имеет аргумент. Аргумент – это переменная, которая позволяет запустить триггер только для определенного действия из класса. Рассмотрим пример:

addmprog 11401 speech 123

Мы добавили мобу скрипт с внумом 11401 на триггер speech с аргументом 123. Триггер speech запустит скрипт, когда персонаж произнесет какую-либо фразу вслух с помощью команды «сказать». Аргумент триггера «123» позволяет нам сразу произвести отбор из всего количества фраз, которые может сказать персонаж. Благодаря аргументу скрипт сработает только тогда, когда персонаж произнесет «123».
Список и описание триггеров и аргументов вы можете найти в соответствующей главе.

1. Работа с редактором скриптов
Пожалуй, пора перейти к объяснению принципов работы скриптов.

1. Компиляция, проверка на ошибки, форматирование

Компиляция (в нашем случае) – процесс перенесения скрипта в некоторую промежуточную, более удобную для обработки форму. Каждый скрипт должен быть уже скомпилирован при выполнении. Если скрипт не скомпилирован – мад выдаст ошибку. Чтобы скомпилировать скрипт, вам нужно зайти в его редактор и ввести «done». Если скрипт не содержит грамматических ошибок, то он скомпилируется.

Скрипт можно проверить на ошибки, не производя попытки его компиляции. Для этого в редакторе надо ввести «check». Если скрипт содержит грамматические ошибки, вам выведется сообщение (при включенной опции wiznet script).

Для улучшения читабельности кода я настоятельно рекомендую форматировать скрипты. Делается это с помощью команды format внутри редактора скриптов.
NB Обратите внимание, что форматируются только те скрипты, которые не содержат грамматических ошибок!

1.2 Исполнение скрипта

Скрипты исполняются построчно, то есть, простым языком говоря, компилятор берет строчку и преобразует ее в машинный код. Заметьте, в этом наши скрипты отличаются от языков программирования типа Pascal, C и т.п.
2. Основные принципы – наследие прогсов

В наших скриптам много чего часто используемого осталось от стандартных прогсов.

2.1 Стандартные моб-команды

В коде моб-скрипта можно указывать команды, которые может делать моб в обычном мире. Для этого достаточно написать команду так, как ее написал бы обычный персонаж. Пример такого скрипта:

ска Привет!

ска Меня зовут Вася!

прыг

Результат будет таков:

Великий Вицлипуцли восклицает: Привет! 

Великий Вицлипуцли восклицает: Меня зовут Вася! 

Великий Вицлипуцли прыгает вокруг, как маленький ребенок.

2.2 Нестандартные команды: мобы, румы и обжекты

Очень часто возникают ситуации, когда вам нужно совершить операции, которые обычному персонажу недоступны. В таких случаях используются специальные команды mob, room и obj соответственно. Сразу маленькое примечание для mob-команд: с помощью команды mob можно делать то же самое, что и без нее. Разница будет в том, что при употреблении слова mob наш моб будет видеть всех (кроме имморталов в инвизе). Без употребления этого слова он увидит только тех, кого он видит со своими детектами. Пример: у моба нет детектов инвиза и скрыта. Мобскрипт:

kill @victim@

Если персонаж запустит скрипт и сам будет невидимым, то результат будет нулевой. Если же скрипт будет таким:
mob kill @victim@

тогда моб нападет на пришедшего. Тут сразу надо заметить, что не все триггера реагируют на невидимые цели (см. раздел про триггера).
Так как комнаты и объекты не могут выполнять команды персонажей, в рум- и обжскриптах соответственно всегда надо пользоваться оператором room и obj.
Рассмотрим наиболее необходимые функции, которые могут использоваться со словами room, mob и obj.
2.2.1 Команды типа echo
Команды типа echo используются для того, чтобы выводить персонажу или группе персонажей какой-либо текст.

· Команда echo выдает сообщение, показываемое всем персонажам в комнате;

· команда echoat (синтаксис: mob echoat <имя>) посылает сообщение только персонажу с данным именем;

· команда echoaround (mob echoaround <имя>) посылает сообщение всем персонажам в комнате, кроме персонажа с данным именем;

· команда echonear (mob echonear <имя1> <имя2>) посылает сообщение всем, кроме указанных двух персонажей;
· командна zecho посылает сообщение всем персонажам в зоне;

· команда gecho посылает сообщение всем персонажам в мире;
· команда asound посылает сообщение во всех комнатах вокруг той комнаты, откуда она производится.

Общий синтаксис команд таков:

mob/room/obj echo <сообщение>

Используя команды этого типа надо быть осторожным, прежде всего пытаться учитывать, слепой персонаж или нет (команда echo выведет слепому такое же сообщение, как и зрячему). Пример на использование команд:
mob echo Вокруг Великого Вицлипуцли начинает сгущаться тьма.

mob echoat Элдерос Скрипт выполняется.

mob echoaround Элдерос А вам ничего не скажу.
mob echonear Элдерос Аэгалил Это увидят все кроме Элдероса и Аэгалила.
mob zecho Слышен глухой колокольный звон.

mob gecho Земная твердь содрогается.

mob asound Быбыбы

В качестве имени обычно используется переменная (см. главу про переменные).

2.2.2 Команды типа transfer
Команды типа transfer предназначены для перемещения персонажей в мире. Они могут работать с двумя типами аргументов – с внумами комнат и с именами персонажей и мобов.

· команда transfer перемещает одного персонажа (моба) в заданную комнату или к заданному мобу (персонажу);

· команда mtransfer перемещает персонажа и его очарованных (призванных) созданий в заданную комнату или к заданному мобу;

· команда gtransfer перемещает всю группу персонажа в заданную комнату или к заданному мобу;
· команда ftransfer перемещает всю цепочку следующий за чаром мобов и чаров, например, если Вася следует за Петей, а Петя – за Машей, то при фтрансфере Маши переместятся все, при фтрансфере Пети – Петя и Вася.

· команда mounttransfer перемещает персонажа, сидящего верхом, и его лошадь, или, если сидят верхом на персонаже, его всадника;

· команда otransfer перемещает объект, лежащий в комнате;

· команда goto используется для перемещения моба, от лица которого выполняется скрипт. Полезной функцией этой команды является то, что если моб дерется, при выполнении команды бой прекращается.

При всех командах типа transfer никакого сообщения персонажам о перемещениях не выводится. Если вы хотите сообщить персонажу, что переместили его или моба – применяйте команды типа echo.

При перемещении персонажа по умолчанию ему выводится текст с описанием комнаты. Если вы хотите переместить персонажа незаметно, напишите в конец команды nolook. При этом персонажу не будет выводиться описание комнаты.
Общий синтаксис команд:

mob transfer <кого перемещать> <куда перемещать> <факультативно: nolook>

Пример:

mob transfer Элдерос 305

mob mtransfer Вицлипуцли Элдерос

mob gtransfer Элдерос 11401 nolook
mob mounttransfer Элдерос 9

mob goto Элдерос
Вместо имени, разумеется, может стоять переменная. В таком случае настоятельно рекомендуется либо брать переменную в кавычки, либо использовать символ «#», чтобы перемещать по уникальному идентефикатору (см. главу про переменные). Иначе, если имя моба окажется из двух слов, вместо места перемещения компилятор увидит второе слово из имени моба и переместит его к самому себе или к другому мобу. Пример:

mob transfer ‘@victim@’ 9037

mob mtransfer #char# 310

2.2.3 Команды удаления предметов и мобов

В этом разделе существует всего 2 команды.

· purge (room purge <имя>) – уничтожает предмет или моба в комнате;

· remove (mob remove <имя или ID персонажа или моба> <внум>) – убирает у персонажа все предметы с заданным внумом, при этом если объект находился внутри контейнера, он не убирается.
Если вы хотите, чтобы объект или моб удалил сам себя, вместо имени используйте команду mob/obj purge self. Примеры:

mob purge ‘трон’

mob remove Элдерос 308

obj purge self
При уничтожении объекта или моба никакого сообщения чарам не выдается! Его надо писать вручную с помощью команд типа echo.
NB Все эхи надо выводить до уничтожения самого себя. Иначе эхи будет выводить некому.
2.2.4 Прочие команды

Также имеется несколько команд, разбивать которые на группы бессмысленно.
· mob cast <заклинание> <цель> <уровень заклинания> - моб (объект, комната) колдует заклинание указанного уровня на цель, чаще всего заклинание ложится не с первого раза, поэтому целесообразно использовать цикл while (смотри раздел про циклы);

· mob oload <внум> <факультативно:цель> <факультативно:слот на цели> - загружает объект с заданным внумом. Если цель не указана, объект загружается в комнату, если указана цель – к цели в инвентарь, если дополнительно указан слот – загружается цели в заданный слот (если слот уже полон – вещь не загружается и выдается баг-репорт). Слоты можно посмотреть в справке, набрав в любом из редакторов «? wear-loc»;

· mob mload <vnum> - в комнату загружается моб с заданным внумом.

· mob removeaff <цель> - убирает с цели все эффекты.
· mob at <vnum> <команда> - совершает команду в комнате с внумом <vnum>. Эта команда не работает в obj-скриптах, вместо нее используйте функцию do_room (см. главу про функции). Если вы хотите указать специфическую команду с mob или room, то повторять слово mob/room не надо! Следующий пример не будет работать:

mob at 123 mob echo Быбыбы
Примеры, написанные правильно:

mob cast ‘левитация’ Элдерос 500

mob oload 308 Элдерос wielded

mob mload 10399
mob at 305 ска Здравствуй, Элдерос!

room at 340 echo Привет, Аэгалил!

При этих командах также не выдается никаких сообщений (только сообщение о результате заклинания).

3. Переменные и их типы

Что такое переменная? Переменное – ссылка на память. В памяти может содержаться определенное значение. Чтобы объявить переменную надо ввести следующую строку кода:

var @<имя переменной>.
Обратите внимание на точку в конце – она обязательна. Имя переменной может состоять из латинских букв, цифр и символа нижнего подчеркивания «_».
В наших скриптах существует неколько типов переменных:

1. Тип integer – целое число.

2. Тип string – строка символов.

3. Тип null – ноль, то есть переменная, которая либо ни имеет типа, либо пуста.

4. Тип mob_data – структура переменных. Хранит в себе параметры моба или чара.

5. Тип room_data – структура переменных. Хранит в себе параметры комнаты.

6. Тип obj_data – структура переменных. Хранит в себе параметры объекта.

7. Тип exit_data – структура переменных. Хранит параметры выхода.

8. Тип array – массив переменных.

При объявлении переменных тип указывать не нужно – он присвоится автоматически при первом присваивании значения. Переменные, которым еще не присвоен тип, имеют значение null.
NB Объявление переменной не является обязательной процедурой! Фактически, переменная создается при первом присваивании ей значения. Однако, советую объявлять переменные для ясности в скрипте. Иначе их будет легко спутать с глобальными переменными. Не объявляйте их только в тех случаях, когда у вас в скрипте катасторфически не хватает строк.
3.1 Операции с переменными
3.1.0 @victim@ и @victim
При использовании переменных их имена имеют два типа записи, как в заголовке этого подпункта. Если вы используете переменную в room-, mob- или obj-команде (включая те mob-команды, которые используются без слова mob), тогда вам необходимо заключать переменную с обоих сторон символами @. В остальных случаях @ ставится только перед именем переменной.
3.1.1 Присваивание значений
Для начала нужно объяснить, как происходит процесс присваивания переменной значения. Оператором присваивания является «:=». В конце каждого присваивания переменной обязательно должна стоять точка. Пример:

var @chislo.

@chislo :=  5 + 5 * 5.

При присваивании переменной значения сначала вычисляется правая часть выражения, потом результат вычисления помещается в переменную, указанную в левой части.
NB Обратите внимание, что переменной с уже определенным типом нельзя присвоить значение другого типа. То есть, выражение «@int := @mob.», где @int имеет тип integer, а @mob – тип mob_data, будет иметь непредсказуемый результат.
При присваивании значения переменной типа string напрямую, необходимо брать строку в одинарные кавычки:

var @str.

@str := ‘Привет!’.

Можно присвоить значение сразу нескольким переменным в одной строке. Для этого следует лишь повторить оператор присваивания необходимое число раз, как в следующем примере:
@var1 := @var2 := … := @var153 := 123+35.
В такой конструкции сначала вычисляется выражение справа (123+35), затем его значение присваивается самой правой переменной (@var153), затем значение переменной @var153 присвоится переменной @var152, которая стоит левее, и так до самого конца.
3.1.2 Доступ к элементам структуры

В скриптах Нерониса имеется три структурных типа переменных: room_data, mob_data и obj_data. Что такое структура? Структура это набор переменных, доступ к которым осуществляется с помощью одного общего имени. Структура может содержать так же вложенные структуры. Доступ к элементам структуры осуществляется с помощью оператора «->». Поясню на примере. Создадим румскрипт и поставим его в комнату на триггер speech с аргументом 123. Триггер speech имеет встроенную переменную @victim@ с типом mob_data (про встроенные переменные смотри ниже). Напишем следующий румскрипт:
room echo @victim@
room echo @victim->nrod@

room echo @victim->ndat@

room echo @victim->level@
room echo @victim->in_room->name@

room echo @victim->in_room->people->level@

При исполнении этого скрипта персонажем Элдерос, находящимся в комнате «Перед лесом» будут получены следующие результаты:

Элдерос

Элдероса

Элдеросу

59

Перед лесом

59

Давайте разбирать подробно.

room echo @victim@

Комната дает эху с содержанием @victim@. @victim@, употребленное без доступа к внутренним переменным, аналогично @victim->nime@ - именительному падежу.
room echo @victim->nrod@

Содержание эхи – родительный падеж имени персонажа @victim. Следующие две строки аналогичны этой, только указывается дательный падеж имени и уровень персонажа.

room echo @victim->in_room->name@
@victim->in_room – элемент структуры mob_data, который имеет тип room_data. Проще говоря, это переменная комнаты, в которой находится персонаж. Так как тип room_data тоже структурный, мы можем получить доступ к следующему уровню и взять на выход имя комнаты.

room echo @victim->in_room->people->level@

Сначала мы получаем доступ к переменной комнаты, из переменной комнаты снова получаем доступ к переменной типа mob_data – части структуры типа room_data, потом получаем уровень моба/персонажа. В данном случае персонажем оказался снова Элдерос, поэтому снова уровень 59. Но, в принципе, там мог бы быть другой персонаж и уровень мог оказаться другим.
Список элементов структуры room_data
	Имя
	Тип
	Описание

	area_maxvnum
	integer
	Максимальный внум зоны, в которой находится комната.

	area_minvnum
	integer
	Минимальный внум зоны, в которой находится комната.

	area_name
	string
	Имя зоны, в которой находится комната.

	area_repopmess
	string
	Сообщение о репопе зоны.

	contents
	obj_data
	Первый предмет в комнате.

	door_north

door_east

door_south

door_west

door_up

door_down
	exit_data
	Возвращает переменную соответствующего выхода.

	down
	room_data
	Комната внизу.

	east
	room_data
	Комната на востоке.

	flags
	-
	Поле флагов, из которого можно выбирать отдельные элементы (см. главу про функции).

	heal_rate
	integer
	Процент восстановления жизни.

	mana_rate
	integer
	Процент восстановления маны.

	name
	string
	Имя комнаты.

	north
	room_data
	Комната на севере.

	people
	mob_data
	Первый персонаж или моб в комнате.

	russector
	string
	Название сектора комнаты (по-русски).

	sector
	string
	Название сектора комнаты (по-английски).

	south
	room_data
	Комната на юге.

	vnum
	integer
	Внум комнаты.

	up
	room_data
	Комната вверху.

	vflags
	-
	Поле флагов, которое содержит в себе флаги виртуальной комнаты.

	weather_flags
	-
	Поле флагов, в котором можно проверить погодные флаги в комнате.

	west
	room_data
	Комната на западе.


Список элементов структуры mob_data
	Имя
	Тип
	Описание

	carrying
	obj_data
	Первый объект в инвентаре моба.

	class
	string
	Профессия персонажа (по-английски).

	clan
	string
	Название клана, в котором стоит персонаж. Если персонаж не состоит в клане – null.

	clanrank
	integer
	Ранг персонажа в клане (цифрой).

	clanrank_str
	string
	Ранг персонажа в клане (строкой – название ранга).

	con
	integer
	Сложение чара.

	dex
	integer
	Ловкость чара.

	deity
deity_nime

deity_nrod

deity_ndat

deity_nvin

deity_ntvo

deity_npre
	string
	Имя божества (по-русски). И соответствующие падежи.

	exp_per_level
	integer
	Количество опыта на левел. Для мобов равно нулю.

	faith
	integer
	Текущее количество очков веры у персонажа.

	fighting
	mob_data
	Персонаж, с которым моб дерется. Если не дерется – null.

	glory
	integer
	Текущее количество очков славы у персонажа.

	gold
	integer
	Количество золотых монет у моба.

	hit
	integer
	Текущее количество здоровья у персонажа.

	hunting
	mob_data
	Персонаж, на которого моб охотится.

	immune
	-
	Поле флагов, из которого можно выбирать отдельные элементы (см. главу про функции).

	id
	integer
	Выдает уникальный номер моба. То же самое, что и #mob#

	int
	integer
	Ум персонажа.

	in_room
	room_data
	Переменная комнаты, в которой находится моб.

	leader
	mob_data
	Тот, за кем персонаж следует.

	level
	integer
	Уровень персонажа.

	long
	string
	Поле long персонажа. Выводится при взгляде на комнату.

	mana
	integer
	Текущее количество маны у персонажа.

	master
	mob_data
	Лидер группы персонажа.

	move
	integer
	Текущее количество шагов у персонажа.

	mount
	mob_data
	Животное, на котором сидит персонаж.

	material_num
	integer
	Номер материала, из которого сделан персонаж.

	max_hit
	integer
	Максимальное количество здоровья у персонажа.

	max_mana
	integer
	Максимальное количество маны у персонажа.

	max_move
	integer
	Максимальное количество шагов у персонажа.

	name
	string
	Имя персонажа.

	next_world_player
	mob_data
	Переменная следующего персонажа в списке «кто».

	nime
	string
	Именительный падеж из поля short.

	nrod
	string
	Родительный падеж из поля short.

	ndat
	string
	Дательный падеж из поля short.

	nvin
	string
	Винительный падеж из поля short.

	ntvo
	string
	Творительный падеж из поля short.

	npre
	string
	Предложный падеж из поля short.

	next_in_room
	mob_data
	Персонаж, следующий за данным персонажем по порядку в комнате.

	pet
	mob_data
	Питомец персонажа (тот, которого он купил в магазине).

	position
	integer
	Возвращает номер позиции, в которой находится моб. Вот номера (можно вводить и названия):

pos_dead
0
pos_mortal
1
pos_incap
2
pos_stunned
3
pos_sleeping
4
pos_resting
5
pos_sitting
6
pos_bashed
7
pos_fighting
8
pos_standing
9


	race
	string
	Название расы (по-английски).

	resist
	-
	Поле флагов, из которого можно выбирать отдельные элементы (см. главу про функции).

	rider
	mob_data
	Персонаж, который сидит верхом на данном мобе.

	rusclass
	string
	Название профессии персонажа (по-русски).

	rusrace
	string
	Название расы персонажа (по-русски).

	sex
	integer
	Пол персонажа (числом), 0 – неопределенный, 1 – мужчина, 2 – женщина.

	short
	string
	То же, что и nime.

	silver
	integer
	Количество серебряных монет у персонажа.

	str
	integer
	Сила персонажа.

	vnum
	integer
	Внум моба.

	vuln
	-
	Поле флагов, из которого можно выбирать отдельные элементы (см. главу про функции).

	wealth
	integer
	Абсолютное богатство персонажа в серебряных монетах. Золотая монета – 100 серебряных.

	weight
	integer
	Вес персонажа. В 10 раз больше того, который выводится персонажу самому.

	wis
	integer
	Мудрость персонажа.


Список элементов структуры obj_data
	Имя
	Тип
	Описание

	carried_by
	mob_data
	Персонаж, который несет этот предмет.

	canwear
	-
	Поле флагов, из которого можно выбирать отдельные элементы (см. главу про функции).

	condition
	integer
	Процент состояния предмета.

	contains
	obj_data
	Первый предмет в контейнере.

	cost
	integer
	Стоимость предмета в серебре.

	extra
	-
	Поле флагов, из которого можно выбирать отдельные элементы (см. главу про функции).

	id
	integer
	Выдает уникальный номер объекта. То же самое, что и #obj#

	in_room
	room_data
	Комната, в которой лежит предмет. Если предмет в инвентаре – null.


	in_obj
	obj_data
	Контейнер, в котором лежит предмет.

	level
	integer
	Уровень предмета.

	long
	string
	Поле long предмета. Показывается, когда он лежит в комнате и чар смотрит без аргумента.

	material
	string
	Материал, из которого сделан объект (по-английски).


	material_num
	integer
	Номер материала, из которого сделан объект.

	name
	string
	Имя объекта из поля name.

	nime
	string
	Именительный падеж объекта.

	nrod
	string
	Родительный падеж объекта.

	ndat
	string
	Дательный падеж объекта.

	nvin
	string
	Винительный падеж объекта.

	ntvo
	string
	Творительный падеж объекта.

	npre
	string
	Предложеный падеж объекта.

	next_content
	obj_data
	Следующий объект в списке – независимо от того, в контейнере ли он, в инвентаре или в комнате.

	owner
	string
	Владелец объекта.

	rusmaterial
	string
	Название материала (по-русски).

	rustype
	string
	Тип объекта из поля type (по-русски).

	short
	string
	То же, что и nime объекта.

	timer
	integer
	Время до распада объекта.

	type
	string
	Тип объекта из поля type (по-английски).

	v0-v4
	разный
	Возвращает значения полей v0-v4 у объекта. Тип значений такой же, как в OLC.

	vnum
	integer
	Внум объекта.

	wear
	-
	Поле флагов, из которого можно выбирать отдельные элементы (см. главу про функции).

	weight
	integer
	Вес объекта. В 10 раз больше того, который выводится персонажу.

	wear_by
	mob_data
	Персонаж, на которого одет объект.


Список переменных структуры exit_data
	Имя
	Тип
	Описание

	affected
	-
	Поле флагов, которое содержит аффекты, которые есть на выходе. Проверять через функцию is_set_flag.

	flags
	-
	Поле флагов, которое содержит флаги, которые есть на выходе. Проверять через функцию is_set_flag. Пример:

 @x := is_set_flag(@room->door_north->flags, exit_closed) ? "за" : "от".
 room echo Дверь на севере @x@крыта.

	key
	integer
	Внум ключа, открывающего дверь.

	name

nime

nrod

ndat

nvin

ntvo

npre
	string
	Имя и падежи выхода.

	to_room
	room_data
	Комната, в которую ведет выход.


3.1.3 Арифметические операции
Немного о математических операциях:

2. Знаки «*» - умножение, «/» - отрицание, «+» - сложение, «-» - вычитание. Работают все эти знаки только для типа integer.
3. Знаки «<» - меньше, «<=» - меньше или равно, «>» - больше, «>=» - больше или равно. Используются для типа integer.
4. Знак «==» - равенство. Проверяет переменные на равенство – используется для любых типов.

5. Знак «!=» - не равно. Используется для любых типов.
6. Знак «-», стоящий перед именем переменной, обозначает унарный минус. В результате получаем отрицательное значение переменной. Только для типа integer.
Чтобы соединить строки, пользуйтесь функциями (см. соответствующий раздел).

NB При сравнивании строк следует заключать строки в кавычки!

3.1.4 Операторы инкремента и декремента

Употребление двух этих операций требует отдельной оговорки. Операторы инкремента и декремента обозначаются знаками «++» и «--» соответственно. Работают они только с типом integer. При инкрементации значение переменной возрастает на единицу, при декрементации значение становится меньше на единицу. То есть, выражение
@x++.

будет эквивалентно выражению
@x := @x + 1.
Существуют две формы употребления операторов инкремента и декремента – префиксная (++@x) и постфиксная (@x++).  Между ними существует качественное различие. При использовании этих операторов в выражении в префиксной форме сначала будет произведена инкрементация, и полученное число будет использоваться в выражении. При использовании постфиксной формы сначала будет посчитано выражение с учетом старого значения переменной, а потом значение переменной увеличится на 1. Поясню на двух простых примерах:
@x := 10.

@y := ++@x.

Переменная @y получит возросшее значение переменной @x – 11.
@x := 10.

@y := @x++.

В этом же случае переменная @y получит старое значение – 10, а потом значение переменной @x увеличится.

3.1.5 Составные операторы присваивания

Помимо основного оператора присваивания, существуют также и составные. Они имеют следующий вид:

+=

-=

*=

/=

Выражение

@x *= 10.

будет аналогично следующему выражению:

@x := @x * 10.
Таким же образом работают и остальные составные операторы присваивания.
NB Слева от основного оператора присваивания должна стоять переменная (не константа). Иначе компилятор выдаст ошибку.
3.2 Встроенные переменные

Каждый триггер имеет свои встроенные переменные. Встроенные переменные – это те переменные, которые имеются в триггере по умолчанию, то есть объявлять их не надо, и каждая из этих переменных уже имеет свое значение. Рассмотрим как пример триггер speech, который имеет переменные @mob (@room, @obj – для каждого типа скриптов свое), @victim (тот, кто сказал фразу) и @argument (произнесенная фраза). Создадим мобпрогс и напишем следующий скрипт:

ска @mob@
ска @victim@

ска @argument@

Зайдем в редактор моба и добавим скрипт с триггером speech и аргументом -1 (аргумент «-1» обозначает то, что скрипт будет запускаться при любой произнесенной фразе). Залоадим моба. Вот результат:
Ты произносишь: Привет! Я учусь скриптам!

Великий Вицлипуцли произносит: Великий Вицлипуцли

Великий Вицлипуцли произносит: Элдерос

Великий Вицлипуцли произносит: Привет! Я учусь скриптам!

Список встроенных переменных для триггеров вы можете посмотреть в справке.

4. Переходы в коде

4.1 Операторы перехода

Операторы перехода позволяют выполнять или не выполнять ту или иную часть кода в зависимости от некоторых параметров. Их, собственно говоря, всего четыре.

4.1.1 Оператор if … then
Это основной оператор перехода. Конструкция с этим оператором имеет следующий вид:

if (<истинностное условие>) then
<блок кода>

endif
Блок кода выполняется только тогда, когда <истинностное условие> имеет значение «истина». (Подробней про истинностное условие смотри в конце главы.)
NB Заключать условие в скобки обязательно!
Как примет возьмем скрипт на триггер speech с аргументом 123:

ска Ты @victim@.

if (is_immortal(@victim)) then
ска Ты бессмертный.

endif
if (is_evil(@victim)) then
ска Ты злой.

endif

if (is_good(@victim)) then

ска Ты добрый.
endif
if (is_neutral(@victim)) then
ска Ты нейтрал.
endif
Функция is_immortal проверяет, является ли персонаж бессмертным, функции is_evil, is_good, is_neutral проверяют, является ли персонаж злым, добрым или нейтралом соответственно. Вот как отреагирует скрипт, если его запустить персонаж Ратхар (злой тролль 51 уровня):

Великий Вицлипуцли произносит: Ты Ратхар.

Великий Вицлипуцли произносит: Ты злой.
Вот как отреагирует скрипт, если его запустит персонаж Элдерос 59 уровня:
Великий Вицлипуцли произносит: Ты Элдерос.

Великий Вицлипуцли произносит: Ты бессмертный.

Великий Вицлипуцли произносит: Ты нейтрал.

Комментарии, думаю, не нужны.

4.1.2 Оператор else
Поставим такую задачу: если чар злой – моб должен атаковать чара, если же он не злой – моб должен снять шляпу. Проще всего подобные конструкции писать с помощью оператора else. Он служит своего рода вспомогательным оператором к оператору if. Вот его синтаксис:
if (<истинностное условие>) then
<блок кода>
else
<блок кода>

endif
Если условие имеет значение «истина», то выполняется блок кода до else, если ложно – после else. В блоках кода также могут быть операторы if и т.п., что часто бывает довольно полезно. Вот пример скрипта:
ска Ты @victim@.

if (is_immortal(@victim)) then
ска Ты бессмертный.

else
ска Ты смертный.
endif
Вот результат для персонажа Ратхар:

Великий Вицлипуцли произносит: Ты Ратхар.

Великий Вицлипуцли произносит: Ты смертный.
Вот для персонажа Элдерос:
Великий Вицлипуцли произносит: Ты Элдерос.

Великий Вицлипуцли произносит: Ты бессмертный.

4.1.3 Оператор elseif
Если нам нужно выбрать один вариант из множества, тогда полезным оказывается оператор elseif. Вот его синтаксис:
if (<истинностное условие>) then
<блок кода>
elseif (<истинностное условие>) then
<блок кода>
elseif (<истинностное условие>) then
<блок кода>
endif
Конструкция выше полностью аналогичная следующей:

if (<истинностное условие>) then
<блок кода>
else
if (<истинностное условие>) then
<блок кода>
else
if (<истинностное условие>) then
<блок кода>
endif
endif
endif
Налицо два преимущества: лучшая читабельность и меньшее количество строк.

NB Всегда лучше вместо двух взаимоисключаемых конструкций с if использовать конструкции с else или с elseif. Дело в том, что если условие верно, то блок кода после else или elseif не выполняется (т.е. состояние персонажа проверяется только один раз). Если же мы имеем две конструкции с if, то состояние персонажа будет проверяться два раза, и на практике часто бывает так, что за время выполнения первого блока кода состояние персонажа изменится, и второй код блока тоже выполнится.

4.1.4 Оператор «… ? … : …»

Оператор ? аналогичен оператору if … else … endif. Это более короткая, но вместе с тем менее функциональная форма. Синтаксис оператора таков:

<истинностное условие> ? <выражение 1> : <выражение 2>
Если условие имеет значение «истина», выполняется выражение 1, иначе – выражение 2. Вот пример использования оператора:
is_npc(@victim) ? @string := ‘моб’ : @string := ‘чар’.
Так как скрипты Нерониса чувствительны к строка, вместо <выражение 1> и <выражение 2> может стоять только одна строка, так что это годится только для коротких выражение. Для более длинных выражений необходимо использовать конструкцию с if … else … endif.
4.2 Использование цикла while
Цикл while позволяет повторить некоторый блок кода несколько раз. Вот его синтаксис:
while (<истинностное условие>) do
<блок кода>

endwhile
Расскажу немного подробнее о том, как работает цикл. При входе в цикл высчитывается условие. Если оно истинно, начинает выполняться блок кода. Когда завершает выполняться блок кода, обработка возвращается к началу цикла и пересчитывает условие (предполагается, что некоторые параметры, от которых зависит истинность условия, изменились), если условие верно, все начинается заново, если неверно – цикл пропускается и начинает обрабатываться код, который написан дальше.

Цикл этот рассчитан на довольно ограниченное количество повторений – что-то около 30. Так что старайтесь уложиться в эти 30 повторений, иначе скрипт будет прерван из-за большого количества действий. Ограничение это можно обойти, если добавить в начало блока кода следующую строку:

delay (0).
На работе скрипта это не скажется никак, однако позволит обмануть интерпретатор. Однако, будьте осторожны! Максимум, что позволяет сделать эта обманка, это в среднем где-то 100 итераций. Если итераций будет больше – рухнет мир. Так что особенно внимательно следите, чтобы ваше условие в нужный момент изменялось на ложное, а также добавляйте дополнительное ограничение на количество повторов!
Вот простейший пример на использование цикла while:
ска Привет!

ска Я умею считать до 10!

var @chislo.

@chislo := 0.

while (@chislo <= 10) do

@chislo := @chislo + 1.

ска @chislo@

endwhile
Можете скопировать и посмотреть результат – моб последовательно скажет числа от 1 до 10.

Оператор break
Оператор break позволяет прервать выполнение цикла while и перейти к следующей строке кода. Как правило, можно всегда обойтись и без него, однако он может сохранить нам пару драгоценных строк и позволить чуть-чуть меньше загружать интерпретатор. Вот пример использования (room-скрипт, триггер speech с аргументом 123):
var @tmp.

@tmp := @room->people.

while (@tmp != NULL) do

if (@tmp->level > @victim->level) then

break

endif
@tmp := @tmp->next_in_room.

endwhile
if (@tmp == NULL) then

room echoat @victim@ Ты тут самый большой!

else
room echoat @victim@ А @tmp@ побольше тебя!
endif
Что делает скрипт? Задача его – посмотреть, есть ли в комнате персонаж, уровень которого больше говорящего. Потом, если такой персонаж находится, то цикл прерывается (ведь считать дальше нам не надо), при этом переменная @tmp сохраняется. Это можно было бы написать по-другому:
var @tmp.

@tmp := @room->people.
vat @char.
while (@tmp != NULL) do

if (@tmp->level > @victim->level) then

@char := @tmp.
endif
@tmp := @tmp->next_in_room.

endwhile
if (@char == NULL) then

room echoat @victim@ Ты тут самый большой!

else
room echoat @victim@ А @char@ побольше тебя!
endif
У метода написания с break для данной задачи есть два преимущества: не нужно выделять память для второй переменной, т.е. скрипт короче и требует меньше памяти; также цикл более экономен, потому что не требуется проходиться по всем персонажам в комнате. Еще между двумя скриптами есть разница: первый выдаст имя первого чара, который выше левом, второй – имя последнего.
Оператор continue
Иногда бывает полезно прервать текущую итерацию цикла, но не прерывать весь цикл. Оператор continue прерывает текущую итерацию и переводит скрипт к проверке условия while’a. Вот пример:
var @i.

var @char.

@i := 10.

@char := @room->people.

while (@i && @char) do

if (is_npc(@char)) then

@char := @char->next_in_room.

continue

endif

@tmp[‘@i’] := @char.
@i--.
@char := next_in_room.

endwhile
Этот скрипт позволяет нам занести в массив первых 10 чаров в комнате. Как правило, continue можно довольно легко заменить условием if … else, однако оно создает лишнюю вложенность, и читабельность скрипта снижается. Про условие цикла и массив см. ниже.
4.3 Логические операции и истинностные таблицы

Я тут везде говорю об истинностных условиях и уже вроде даже упоминал сложные условия. Пришло время разобраться в них поподробней.

Истинностное условие – выражение, которое может иметь на выходе два значения: «истина» (И) и «ложь» (Л). Но, так как специального типа, который может принимать только два значения, у нас в скриптах нет, принято соглашение, что ложью считаются следующие вещи:

· значение переменной null;
· значение переменной 0;
· значение переменной ‘’, то есть пустая строка.

· переменная содержит not-valid руму, моб, обж или выход (т.е. рума, выход или обж были удалены или моб убит/спуржен).

· переменная содержит пустой массив (без ключей).

· если переменная содержит неизвестный тип данных.

Все остальные значения считаются истиной. Соответственно, все операции и функции, которые подразумевают на выходе И или Л (например, «>»), выдают вовсе не их, а нули или не нули. Однако для простоты я сохраню обозначения Л для нуля и И для не нуля.
Теперь об истинностных таблицах. Истинностная таблица позволяет узнать результат логической операции, основываясь на ее компонентах. Так как у нас  логические операции только унарные и бинарные (1 и 2 аргумента соответственно), то таблица будет иметь вид обычной двумерной таблицы, так что с этим все просто. В столбике слева – аргумент слева от знака операции, в столбиках сверху (если они есть) – аргумент справа, если он есть. Вот список логических операций
· Логическая операция «отрицание» - обоначается знаком «!». Ставится непосредственно перед переменной и превращает соответственно ложь в правду а правду в ложь (. Часто бывает полезен, когда надо отрицать какую-либо функцию. Таблица:

	И
	Л

	Л
	И


То есть, если у нас аргументом будет «истина», общее выражение будет иметь значение «ложь». Пример:

if (!is_npc(@victim)) then
Функция  is_npc определяет, является ли персонаж с типом mob_data мобом. Соответственно отрицание этой проверки будет являться проверкой того, является ли персонаж чаром.

· Логическая операция «и» - обозначается знаком «&&». Истинностная таблица:

	
	И
	Л

	И
	И
	Л

	Л
	Л
	Л


Общее выражение на выходе будет иметь значение «истина», только если оба аргумента имеют значение «истина». Пример:
if ((2 > 1) && (@victim->name == ‘Элдерос’)) then
· Логическая операция «или» - обозначается знаком «||». таблица:

	
	И
	Л

	И
	И
	И

	Л
	И
	Л


На выходе получаем «ложь», только если оба аргумента ложны.
Также у нас в скриптах есть еще две операции – побитовое «и» (&) и побитовое «или» (|). В логических операциях они ничем не отличаются от обычных «и» и «или». Но вообще, я настоятельно рекомендую использовать логический (двойной) вариант, а про побитовый забыть по крайней мере до тех пор, пока вы сами не сможете написать такой же мануал по скриптам.
4.4 Приоритет операций и расстановка скобок

Приоритет операций, как и следовало ожидать, различен. Как происходит выбор текущей операции? Компилятор проходится по строке и ищет самую вложенную скобку. Найдя ее, он начинает обработку выражения внутри этой скобки. Идя слева по порядку, он ищет самую приоритетную операцию. Найдя, вычисляет ее результат. Потом снова начинает ту же проверку. Когда в скобке остается одно число – скобки опускаются. 
Вот список операций – чем выше, тем выше приоритет:

++, --, -(унарный минус)

!
*, /
+, -
<, >, >=, <=
!=, ==
&
|
&&
||
?:
:=, +=, -=, *=, /=
Круглые скобки повышают приоритет операций – сначала будут выполнены операции в скобках. Использование других скобок не допускается. Лишние скобки не замедляют процесса вычисления. Однако чрезмерная нагрузка выражения скобками сильно снижает читабельность скрипта. Так что, старайтесь не ставить лишних скобок.
4.5 Ограничения на скрипты и вызов других скриптов

Для сокращения памяти, которая тратится на скрипты, у нас действуют некоторые ограничения. В одном скрипте может быть не более 75 строчек и не более 4000 знаков. Что делать, когда нужно больше места? Вызвать другой скрипт. Вызов другого скрипта, помимо продолжения предыдущего, имеет кучу применений, которые вы найдете сами.
Вызов другого скрипта осуществляется с помощью команды call. Синтаксис:

call <vnum>.

Вызывать можно только скрипты того же типа, что и сам скрипт, из которого мы вызываем другой. При вызове сохраняются все переменные со всеми значениями. Вместо <vnum> может стоять любое выражение типа integer, например 2543+5748-@int. Точка в конце (по последним слухам) не обязательно, хотя я ставлю для верности.
NB Существует также ограничение на вызов скриптов. Можно вызывать не более 3 вложенных событий. Однако его можно обойти с помощью специальной функции set_max_calls <число>. Для использования этой функции скрипту необходим уровень security 1.
5. Работа с массивами

Работу с массивами я решил вынести из раздела с переменными, так как она устроена несколько сложнее.
7. Что такое массив

Массив в скриптах Нерониса – собрание переменных, объединенных общим названием (и, собственно говоря, ничем, кроме названия). Переменные внутри массива (=элементы массива) могут иметь разные типы. Каждый элемент массива имеет свой индекс. Чтобы получить доступ к элементу массива, надо воспользоваться следующим типом записи:

@<имя массива>[<индекс элемента>]
Однако, следует учесть, что в mob-, obj- и room-командах можно использовать только обычные переменные. Т.е. нельзя написать

room echo @day[1]@ 

Для использования элемента массива в таких случаях надо сначала присвоить его значение промежуточной переменной:

@tmp := @day[1].

room echo @tmp@

Создавать элементы массива можно также с помощью функции array(см. главу про функции).
5.1 Индексация массивов

Индексация массивов устроена по строкам, т.е. массивы ассоциативные. Например, индексом можно указать имя персонажа. Если вы в качестве индекса массива укажите цифру, компилятор будет воспринимать ее как строку.
5.2 Перебор элементов массива

Довольно часто возникает необходимость перебрать все элементы массива.

8. Перебор с помощью функций

Основной способ переборки массива – с помощью функций reset и each. Дело в том, что у каждого массива существует скрытый от глаз пользователя указатель на текущий элемент массива.
NB Для тех кто работал с Pascal: система индексации наших массивов устроена почти так же, как работа с текстовыми файлами в Pascal’е.
Функция reset служит для того, чтобы поставить указатель на первый элемент массива. Функция each выдает значение текущего элемента массива и переводит указатель на следующий элемент массива. Как только функция each выдает null, мы можем считать, что мы вышли за пределы массива (хотя это, к сожалению, не всегда так). В этом случае следует закончить процедуру переборки массива. Давайте рассмотрим пример (можете использовать его в качестве шаблона). Допустим, в глобальной переменной мы ведем счетчик посещения каким-то персонажем мира в определенный день недели, и нам необходимо написать скрипт, который бы выдавал результаты этой слежки. Вот, что у нас уже есть в глобальной переменной:
@washere[‘понедельник’] := 45.

@washere[‘вторник’] := 27.

@washere[‘среда’] := 64.
@washere[‘четверг’] := 39.
@washere[‘пятница’] := 42.
@washere[‘суббота’] := 7.
@washere[‘воскресенье’] := 13.
В скрипте нам надо будет перебрать все элементы этого массива и выдать результаты. Для простоты будем обозначать порядковый номер дня недели цифрой и через тире ставить результат.
var @day.
@day := 1.
var @element.

reset(@washere).
@element := each(@washere).
while (!is_nullvar(@element)) do
room echo В @day@-й день недели - @element@.
@day := @day + 1.

@element := each(@washere).

endwhile
В третьей строке мы объявили промежуточную переменную, в которой будут храниться данные из текущего элемента массива. В четвертой – установили счетчик массива на первый элемент. В пятой строке в переменную @element перенесено значение текущего, то есть первого, элемента массива (чтобы она не имела значения null), а счетчик массива переставлен на позицию второго элемента. Далее заводим цикл while, условием выхода из которого будет значение null переменной @element. Выводим эху с первым элементом массива, затем увеличиваем счетчик дней, в переменную @element заносим значение следующего элемента массива, а счетчик массива переводим на следующий элемент. Все. Когда цикл while будет повторяться шестой раз, счетчик массива передвинется на несуществующий элемент, следовательно в седьмой раз повторения цикла в переменную @element будет занесено значение null, и цикл прервется.

Можно записать это короче (можно использовать в качестве шаблона):

var @day.

@day := 1.
var @element.

reset(@washere).
while (!is_nullvar(@element := each(@washere))) do

room echo В @day@-й день недели - @element@.

@day := @day + 1.

endwhile
В это случае перенесение текущего элемента в промежуточную переменную будет делаться прямо при проверке цикла. Соответственно, не надо присваивать значение до цикла и внутри цикла.
5.2.2 Перебор с помощью числовой индексации

Может сложиться ситуация, когда вы захотите управлять порядком переборки массива. В этом случае перебирать с помощью функций each и reset не удастся. В этом случае можно будет применить частный вариант индексации – индексацию по числам.

NB Такой способ индексации менее эффективный, поэтому, когда можно обойтись без него – обходитесь без него.
Вот пример скрипта с индексацией по числам:

@washere[1] := ‘Фалагар’.
@washere[2] := ‘Аэгалил’.
@washere[3] := ‘Кетан’.
@washere[4] := ‘Элдерос’.
@washere[5] := ‘Кзар’.
@washere[6] := ‘Глорфи’.
var @rand_integer.
var @rand_element.
@i := 0.
while ((@rand_integer := random(7)) && @i < 30) do

@i++.
@rand_element := @washere[@rand_integer].

room echo @rand_element@

endwhile
Давайте разбираться в скрипте. Сначала мы проиндексировали массив по числам и присвоили элементам соответствующие значения. Дальше будут использоваться две переменные: @rand_integer для выбора индекса и @rand_element – промежуточная переменная для элемента с нужным индексом. В условии цикла while стоит выражение, в котором мы выбираем индекс с помощью функции random, которая выдает случайное число. Рано или поздно функция выдаст 0 и условие станет неверным; если итераций будет слишком много, тогда переменная @i обеспечит нам выход из цикла после 30-й операции. Итого, этот скрипт будет выдавать случайный элемент из заданного массива. Вместо случайности можно подставить любую другую формулу, что иногда может быть полезным.
6. Функции и константы
6.1 Функции
Тут я перечислю функции и их аргументы. С комментариями, разумеется. Узнать набор функций можно, набрав ? functions в редакторе скрипта. В скобках у функции указаны аргументы. Аргументы указываются через запятую. Здесь я после каждой переменной после двоеточия пишу тип переменной, которая дается на вход функции. Вместо переменной функции можно давать на вход константу.
Чтобы запустить функцию, нужно просто набрать ее имя и в скобках указать параметры, от которых она должна работать. Также функцию можно писать справа от знака присваивания или внутри скобок условия. В конце следует ставить точку. Примеры:

@newmob := mobhere(1201).

@charnewskill := 10 + getskill(@victim, ‘топоры’).
set_owner(@obj, ‘@victim->name’).
if (!carries(@victim, 1234)) then
while ((@victim->in_room->vnum == 9)&&(!is_lost(@victim))) do

О типе boolean в функциях

Типа boolean (имеет два значение – Истина и Ложь), как уже говорилось выше, у нас в скриптах нет (см. раздел про истинностные условия). Однако описание функций без прибегания к нему становится очень сложным. В таблице ниже функции, где в выдаваемом типе стоит boolean, там функция выдает integer. Просто логически она в 99% случаев используется как «И/Л».

Если вместо выдаваемого типа стоит прочерк – это опять же не значит, что функция ничего не выдает. Выдает, но я, во-первых, не знаю что (скорее всего, integer), а во-вторых, мне в голову не может прийти как ее использовать в выражениях и истинностных условиях. Это мое сугубо личное мнение, так что вы вольны поступать, как вздумаете. Каждая функция что-нибудь да возвращает.
	Функция
	Выдаваемый тип
	Описание

	add_gold(@amount:integer)
	-
	Для моб-скриптов. Добавляет мобу @amount золота.

	add_money( @mob:mob_data,

@gold:integer,

@silver:integer)
	-
	Добавляет мобу @mob @gold золота и @silver серебра. Значения могут быть и отрицательными.

	add_silver(@amount:integer)
	-
	Аналогична add_gold, только добавляет серебро.

	array(@var1 := ‘значение1’, @var2 := ‘значение2’ …)
	-
	Функция для задавания значений элементам массива всем сразу в одной строке. Переменной @var1 присваивается значение «значение1» и т.п.

	badname(@ch:mob_data)
	boolean
	Функция для генерации. Проверяет, не плохое ли имя у @ch.

	can_see(@ch:mob_data, @victim:mob_data)
	boolean
	Проверяет, может ли @ch видеть @victim. Если видит – выдает 1, если нет – 0.

	carries(@ch:mob_data, @vnum:integer)
	obj_data
	Проверяет, несет ли в инвентаре @ch объект с внумом @vnum. Если несет, тогда возвращает переменную этого объекта.

	count_obj(@ch:mob_data, @vnum:integer, @int:integer)
	integer
	Выдает количество объектов с внумом @vnum у чара @ch. Если @int равно 0, то в контейнерах не ищется, иначе – ищется.

	damage(@ch:mob_data, @int:integer)
	-
	Отнимает у @ch @int очков здоровья.

	do_room(@room:room_data, “@str:string”).
	-
	Делает команду @str от имени комнаты @room. Функция полезна, так как инвентарь room-команд больше, чем obj-команд, например. Ну или еще какие действия можно прикрутить. В строке @str указывать слово room не надо. Чтобы получить переменную комнаты, следует пользоваться функцией is_room. Точка в конце обязательна! Пример:

do_room(@room, “echo Элдерос 59 лева”).

	door_close( @door:exit_data)
door_lock(@door:exit_data)
door_open( @door:exit_data)
door_unlock( @door:exit_data)
	integer
	Совершают соответствующие действия с дверью. Если не удается совершить действие, возвращается 0.
NB: Если вы пытаетесь закрыть уже закрытую дверь, например, то результат будет удачный, так как результат достигнут – дверь закрыта :)

	each(@a:array)
	Тип элемента в массиве
	Функция возвращает элемента массива и переводит счетчик индексации массива на следующий элемент. Подробнее см. раздел про массивы.

	gain_exp(@ch:mob_data, @int:integer)
	-
	Начисляет персонажу @int опыта. Для использования этой функции скрипту необходим secure 59.

	gain_glory(@ch:mob_data, @int:integer)
	-
	Начисляет персонажу @int славы. Для использования этой функции скрипту необходим secure 59.

	gain_faith(@ch:mob_data, @int:integer)
	-
	Начисляет персонажу @int веры. Для использования этой функции скрипту необходим secure 59.

	get_char_room( @room:room_data или @mob:mob_data)
	mob_data
	Если в качестве аргумента указано комната, возвращает первого чара в комнате. Если моб – первого чара по списку после этого моба.

	get_char_world(@name:string, @int:integer)
	mob_data
	Выдает чара (не моба) с именем @name. Если @int равно нулю, ищется полное совпадение имени, иначе ищется имя с префиксом @name. Если функция не находит чара, она выдает null.

	get_door( @room:room_data, @dir:integer)
	exit_data
	Возвращает переменную выхода из комнаты @room в направление @dir.

	get_equip(@ch:mob_data, @where:string)
	obj_data
	Выдает объект, одетый у @ch на слот @where. Узнать список слотов можно, набрав ? wear-loc в любом из редакторов. Если слот пуст, выдает null.

	get_material(@int:integer, «@what:string»)
	@what->type
	Выдает, какое-то свойство материала. @int – это @obj->material_num. @what можно выбирать из следующего набора:

name – имя(строка);
rusname – русское имя(строка)

type – тип (строка)

rustype – тип по-русски (строка)

weight – вес (число)

cost – стоимость (число)

durable – прочность (число)

pierce – защита от укола (число)

slash – от рубки(число)

bash – от удара (число)

exotic – от экзота (число)

	get_random_room (@ch:mob_data)
	room_data
	Возвращает случайную комнату в мире, куда может попасть @ch (кроме ловушек и т.п.).

	get_skill(@ch:mob_data, @skill:string)
	integer
	Возвращает процент умения @skill у @ch. Если умения нет, возвращает 0.

	get_sn(@string)
	integer
	Возвращает номер заклинания или умения.

	grpsize(@ch:mob_data)
	integer
	Возвращает количество мобов в группе @ch, которые при этом находятся с ним в одной комнате.

	is_affected(@ch:mob_data, @affect:string)
	boolean
	Проверяет, наложен ли на @ch эффект @affect. Список эффектов можно посмотреть, набрав ? affect в любом редакторе.

	is_evil(@ch:mob_data)
	boolean
	Проверяет, является ли @ch злым.

	is_good(@ch:mob_data)
	boolean
	Проверяет, является ли @ch добрым.

	is_immortal(@ch:mob_data)
	boolean
	Проверяет, является ли @ch имморталом.

	is_killer(@ch:mob_data)
	boolean
	Проверяет, является ли @ch убийцей.

	is_lost(@ch:mob_data)
	boolean
	Проверяет, в лосте ли чар.

	is_master(@ch:mob_data)
	boolean
	Проверяет, является ли @ch лидером группы.

	is_name(@name:string, @allnames:string)
	boolean
	@allnames рассматривается как

список имен и этом списке ищется имя, которое начинается на первую строку. Разделителем для имен в @allnames является пробел.

	is_neutral(@ch:mob_data)
	boolean
	Проверяет, является ли @ch нейтралом.

	is_newb_nsrca(@ch:mob_data)
	boolean
	Функция для генерации. Видимо, проверяет, сгенерился ньюб или нет.

	is_npc(@ch:mob_data)
	boolean
	Проверяет, является ли @ch мобом. Отрицание позволяет проверить, является ли @ch чаром.

	is_nullvar(@var)
	boolean
	Проверяет значение переменной @var. Бывает полезно, если вы не хотите мучаться, с кавычками сравнивать или без кавычек.

	is_number(@arg:string)
	boolean
	Проверяет, является ли @arg числом.

	is_room(@vnum:integer)
	room_data
	Выдает комнату с внумом @vnum. Если такой комнаты не существует, то выдается null.

	is_same_group( @ch1:mob_data, @ch2:mob_data)
	boolean
	Проверяет, находятся ли чары в одной группе.

	is_set_flag(@where:поле

флагов, @what:string)
	boolean
	Проверяет, есть ли в поле флагов @where флаг @what. Поля есть следующие:

@room->flags – флаги комнаты;

@obj->extra
@obj->canwear
@obj->wear
@mob->resist
@mob->vuln
@mob->immune
Вот примеры:

is_set_flag(@victim->resist, resist_magic)
is_set_flag(@obj->extra, extra_glow)
Если да – выдает 1, если нет – 0.

	is_sitting(@ch:mob_data)
	boolean
	Проверяет позицию @ch, если тот сидит – возвращает 1 (истину), иначе – 0 (ложь).

	is_sleeping(@ch:mob_data)
	boolean
	Проверяет, спит ли @ch.

	is_thief(@ch:mob_data)
	boolean
	Проверяет, является ли @ch вором (имеет флаг thief).

	manadamage( @ch:mob_data, @int:integer)
	-
	Отнимает у моба @ch @int маны.

	mob_evented(@vnum:integer)
	integer
	Выдает число одновременно запущенных скриптов с данным внумом. Считает также тот скрипт, из которого запущена эта функция. Для mob-скриптов.

	mobexist(@vnum:integer)
	mob_data
	Ищет в мире моба с указанным внумом. Если моб есть, возвращает его переменную.

	mobhere(@vnum:integer,

фак-но:@where:integer)
	mob_data
	Проверяет, находится ли моб с внумом @vnum в комнате. Если моба с таким внумом нет, тогда возвращает null. Если указано @where, то ищет моба в комнате с внумом @where.

	mobs(@room:room_data,

фак-но: @vnum:integer)
	integer
	Возвращает количество мобов (фак-но: с заданным внумом) в комнате (только npc).

	movesdamage( @ch:mob_data, @dam:integer)
	-
	Уменьшает текущее количество мувов у @ch на @dam.

	objexist(@vnum:integer)
	obj_data
	Ищет в мире обжект с указанным внумом. Если он есть, возвращаего его переменную.

	order(@var:mob_data или obj_data)
	integer
	Возвращает порядковый номер моба в комнате среди мобов того же внума, если @x – моб и номер объекта в комнате для лежащих на земле либо номер в инвентаре для находящихся в инвентаре. Может быть полезно если триггер должен сработать только на одном находящемся в комнате мобе либо объекте. Нумерация начинается с нуля.

	obj_evented(@vnum:integer)
	integer
	Аналогична mob_evented.

	obj_to_char(@obj:obj_data, @ch:mob_data)
	integer
	Перемещает объект @obj в инвентарь к чару @ch. Если переместить нельзя, возвращает 0.

	obj_to_obj(@obj:obj_data, @cont:obj_data)
	integer
	Перемещает объект @obj внутрь контейнера @cont. Нельзя засунуть контейнер в контейнер и объект с флагом noput. Если переместить нельзя, возвращает 0.

	obj_to_room( @obj:obj_data, @room:room_data)
	integer
	Перемещает объект @obj  в комнату @room. Если переместить нельзя, возвращает 0.

	objhere(@vnum:integer, фак-но:@where:integer)
	obj_data
	Выдает объект с внумом @vnum, лежащий в комнате. Если такого нет, выдает null. Если указано @where, то ищет объект в комнате с внумом @where.

	people(@room:room_data)
	integer
	Возвращает количество мобов в комнате (npc и pc).

	players(@room:room_data)
	integer
	Возвращает количество игроков в комнате (только pc).

	random(@int:integer)
	integer
	Выдает случайное число из промежутка [0;@int-1], то есть от нуля до @int – 1 включительно.

	reset (@a:array)
	-
	Возвращает счетчик индексации массива на первый элемент. См. раздел про массивы.

	room_evented(@vnum:integer)
	integer
	Аналогична mob_evented.

	set_owner(@obj:obj_data, @name:string)
	-
	Присваивает объекту @obj владельца @name. Имя владельца писать с большой буквы.

	set_timer(@obj:obj_data, @time:integer)
	-
	Ставит на объект таймер. Через @time секунд объект распадается.

	str_prefix(@prefix:string, @string:string)
	boolean
	Проверяет, является ли строка @prefix префиксом к @string.

	strcat(@str1:string, @str2:string, …)
	string
	Возвращает строку, которая состоит из объединенных по порядку строк @str1, @str2 и т.д.

	strlen(@str:string)
	integer
	Возвращает длину строки @str.

	strpos(@str:string, @substring:string, 
фак-но:@offset:integer)
	integer
	Ищет подстроку @substring в основной строке @string. Если находит строку – возвращает индекс буквы, с которой начинается @substring. Если не находит – возвращает null.
Примечание: индексация букв в строке начинается с нуля: 0, 1, 2 …, n.

	substr(@string:string, @start:integer,

фак-но: @end:integer)
	string
	Возвращает подстроку строки @string начиная с буквы с индексом @start и заканчивая буквой с индексом @end. Если переменная @end не указана, вырезает до конца строки @string.

	wait_state(@ch:mob_data, @time:integer)
	-
	Добавляет чару @ch лаг на @time секунд.

	wears(@ch:mob_data, @vnum:integer)
	obj_data
	Проверяет, одет ли на @ch предмет с внумом @vnum. Если одет, тогда возвращает переменную этого объекта.

	world_players
	mob_data
	Возвращает переменную первого персонажа в списке «кто» (кроме имморталов в визинвизе).


6.2 Константы

Список констант, видимо, будет неполным, пока что. Константы – переменные, которые не меняют своих значений и универсальны (в нашем случае). Их можно использовать вместо соответствующих чисел. Если в таблице в конце константы стоит нижнее подчеркивание, это значит, что это префикс, и чтобы получить название констаны, надо добавить еще название флага. Например, exit_closed.
	Константа
	Значение

	int-данные (проверяются обычно)

	dir_north
	0

	dir_east
	1

	dir_south
	2

	dir_west
	3

	dir_up
	4

	dir_down
	5

	decl_
	Род у предметов.

	item_
	Типы предметов.

	sex_
	Пол у мобов.

	size_
	Размеры.

	int64 (проверяются по is_set_flag)

	act_
	Флаги из поля act на мобе или на трупе.

	armor_flag_
	Флаги на броне.

	charge_
	Тип зарядов (арбалет, праща, лук).

	cont_
	Флаги контейнеров.

	dcont_
	Флаги контейнеров для жидкости.

	exit_
	Флаги выхода.

	exit_aff_
	Эффекты на выходе.

	form_
	Флаги из поля form на мобе или на трупе.

	furn_
	Флаги мебели.

	off_
	Флаги из поля off на мобе или на трупе.

	parts_
	Флаги из поля parts на мобе или на трупе.

	portal_
	Флаги портала.

	weapon_
	Тип оружия.

	weapon_flag_
	Флаги на оружии (морозное и т.п.)

	weather_type_
	Тип погоды (для предметов с типом «weather»).


7. Триггера

Справочный раздел. Для каждого типа скриптов есть свои триггера. Если триггер должен срабатывать всегда, то чаще всего аргументом может быть -1 или all. Сейчас перечислю триггера:

7.1 room-триггера
Для всех общая переменная @room – комната, которая выполняет скрипт.
	Триггер
	Описание
	Аргумент
	Встроенные переменные

	buy
	Срабатывает при покупке предмета.
	Число – внум купленного предмета.
	@victim – кто купил;

@vobj – купленный предмет.

	cast
	Срабатывает при касте заклинания.
	Строка – имя заклинания.
	@who - кто кастит
@sn - номер закла
@engname - английское имя закла
@rusname - русское имя закла
@argument - все, что идет после названия закла
@victim - на кого кастится закл
@vobj - предмет, на который кастится закл (для некоторых заклов, которые хотя
и кастятся на предмет, этоа переменная не устанавливается по тех.
причинам).
@dirname - название направления, в котором кастится закл
@revdirname - название направления, обратного dirname.
@toroom - внум комнаты, в направлении которой было колдунство.

	command
	Срабатывает при вводе существующей команды, команда не исполняется. Чтобы команда исполнилась, надо внутри скрипта сделать форс ее сделать.
	Строка – существующая команда.
	@victim – кто ввел команду;

@command – что за команда введена;

@argument – аргумент команды, которую ввел чар.

	death
	Срабатывает на смерть моба в руме.
	Число – внум моба.
	@victim – кто убил.

	delay
	не работает
	
	

	drop
	Срабатывает, когда кто-то бросил объект.
	Число – внум объекта.
	@victim – кто бросил;

@dropped – брошенный объект.

	eat
	Срабатывает при съедании обжекта.
	Число – внум обжекта.
	@victim – кто съел;

@vobj – что съел.

	enterall
	Срабатывает при входе в комнату. При этом вроде реагирует на иммов и невидимых всяких.
	Число – вероятность.
	@victim – кто пришел.

	exall
	Срабатывает, когда чар пытается выйти из румы. Чтобы чар вышел, ему надо сделать форс на направление внутри скрипта.
	Число – направление.

0 – север, 1 – восток, 2 – юг, 3 – запад, 4 – верх, 5 – низ.
	@victim – кто уходит;

@dir – куда идет (числом);

@dirname – куда идет (строкой);
@revdirname – выход, противоположный dirname;
@toroom – внум комнаты, в которую он попадет.

	fight
	Срабатывает во время боя каждый батлтик. Причем срабатывает после атак того, кто начал драку, до других атак.
	Число от 1 до 100. Вероятность.
	@victim – кто запустил.

	flee
	Срабатывает при бегстве чара из комнаты.
	Число – вероятность.
	@victim – кто сбежал;

@dir – куда идет (числом);

@dirname – куда идет (строкой);

@revdirname – выход, противоположный dirname;

@wasvnum - vnum комнаты, где персонаж был;
@wasroom (room_data) - комната, где персонаж был;

	get
	Срабатывает при подбирании предмета.
	Число – внум подобранного предмета.
	@victim – кто купил;

@vobj – подобранный предмет.

	give
	Срабатывает, когда кто-то кому-то что-то дает.
	Число – внум обжекта, который дали.
	@victim – кто дал;

@vobj – что дал.

	greet
	Срабатывает, когда чар появляется к комнате (в т.ч. и из портала, на трансфер не работает). Работает только один скрипт с таким триггером!
	Число – вероятность.
	@victim – кто пришел.

	look
	Срабатывает, когда чар смотрит на какой-то объект.
	Число – внум объекта.
	@victim – кто смотрит;

@vobj – объект, на который смотрят.

	new_cmd
	Срабатывает при вводе несуществующей команды. Своего рода эмуляция оной.
	Строка – несуществующая команда.
	@victim – кто ввел команду;

@command – что за команда введена;

@argument – аргумент команды, которую ввел чар.

	open
	Срабатывает при открывании контейнера в руме.
	Число – внум контейнера.
	@victim – кто открыл;

@vobj – что открыл.

	post_cmd
	Срабатывает после ввода существующей команды, команда исполняется.
	Строка – существующая команда.
	@victim – кто ввел команду;

@command – что за команда введена;

@argument – аргумент команды, которую ввел чар.

	random
	Срабатывает каждый тик.
	Число – вероятность.
	Нет.

	sit

rest

sleep

stand
	Срабатывают при попытки сесть, отдыхать, спать и встать.
	Число – вероятность.
	@victim – кто сел и т.п.

	smb_death
	Срабатывает при чьей-то смерти в руме.
	Число – вероятность.
	@victim – кого убили.

	social
	Срабатывает при исполнении какого-то социала.
	Строка – социал.
	@victim – на кого направлен социал;

@who – кто делал социал.

	speech
	Срабатывает, когда чар произносит фразу.
	Строка – сказанная фраза.
	@victim – кто сказал;

@argument – что сказал.

	update
	Срабатывает при репопе зоны.
	Число – вероятность.
	Нет.

	wear
	Срабатывает при одевании объекта.
	Число – внум одеваемого объекта.
	@victim – кто купил;

@vobj – одетый предмет.


С рум-триггерами все.
7.2 mob-триггера

Для всех общая переменная @mob – моб, который делает скрипт.
	bribe
	Срабатывет, когда чар дает мобу определенное количество денег.
	Число – стоимость денег (в серебре).
	@victim – кто дает деньги;

@amount – сколько дал.

	cast
	Срабатывает при касте заклинания.
	Строка – имя заклинания.
	@who - кто кастит
@sn - номер закла
@engname - английское имя закла
@rusname - русское имя закла
@argument - все, что идет после названия закла
@victim - на кого кастится закл
@vobj - предмет, на который кастится закл (для некоторых заклов, которые хотя
и кастятся на предмет, этоа переменная не устанавливается по тех.
причинам).
@dirname - название направления, в котором кастится закл
@revdirname - название направления, обратного dirname.
@toroom - внум комнаты, в направлении которой было колдунство.

	death
	Срабатывает при смерти моба. Причем срабатывает в момент непосредственно до смерти. Если нужно ставить после смерти – надо ставить в начало скрипта delay(0).
	Число – вероятность.
	@victim – кто убил.

	delay
	не работает
	
	

	enter
	Срабатывает при входе в комнату.
	Число – вероятность.
	@victim – кто пришел.

	enterall
	Аналогичен enter, только срабатывает на невидимых для моба.
	
	

	entry
	?
	
	

	exall
	Аналогично exit, только срабатывает на всех, даже невидимых.
	
	

	exit
	Срабатывает, когда чар пытается уйти из комнаты с мобом. Чтобы чар ушел, надо пофорсить его скриптом.
	Число – направление.

0 – север, 1 – восток, 2 – юг, 3 – запад, 4 – верх, 5 – низ.
	@victim – кто уходит;

@dir – куда идет (числом);

@dirname – куда идет (строкой);

@revdirname – выход, противоположный dirname;
@toroom – внум комнаты, в которую он попадет.

	fight
	Срабатывает каждый батлтик.
	Число – вероятность.
	@victim – кого моб бьет.

	flee
	Срабатывает при бегстве чара из комнаты, где стоит моб.
	Число – вероятность.
	@victim – кто сбежал;

@dir – куда идет (числом);

@dirname – куда идет (строкой);

@revdirname – выход, противоположный dirname;
@wasvnum - vnum комнаты, где персонаж был;
@wasroom (room_data) - комната, где персонаж был.

	give
	Срабатывает, когда мобу дают предмет.
	Число – внум предмета, который дают.
	@victim – кто дал;

@vobj – что дал.

	grall
	То же самое, что greet, только работает на тех, кого моб не видит.
	
	

	greet
	Срабатывает, когда чар заходит в комнату с мобом. Работает только один скрипт с этим триггером.
	Число – вероятность.
	@victim – кто пришел.

	hpcnt
	Срабатывает при изменении жизни моба.
	Число – процент жизни, ниже которого срабатывает скрипт.
	?

	kill
	Срабатывает, когда кто-то нападает на моба.
	Число – вероятность.
	@victim – кто напал.

	look
	Срабатывает, когда на моба глядят.
	Число – вероятность.
	@victim – кто посмотрел.

	random
	Срабатывает где-то каждые 6 секунд.
	Число – вероятность.
	Нет.

	smb_death
	Срабатывает при чьей-то смерти в руме.
	Число – вероятность.
	@victim – кого убили.

	social
	Срабатывает при исполнении какого-то социала.
	Строка – социал.
	@victim – на кого направлен социал;

@who – кто делал социал.

	speech
	Срабатывает, когда чар произносит фразу.
	Строка – сказанная фраза.
	@victim – кто сказал;

@argument – что сказал.

	update
	Срабатывает при репопе зоны.
	Число – вероятность.
	Нет.


7.3 obj-триггера
Общая для всех триггеров переменная - @obj.  Скрипты на обжектах не работают, когда обжект внутри контейнера.
	Триггер
	Описание триггера
	Аргумент
	Переменные

	buy
	Срабатывает на покупку обжекта.
	Число – вероятность.
	@victim – кто купил;

@vobj – что купил.

	cast
	Срабатывает при касте заклинания.
	Строка – имя заклинания.
	@who - кто кастит
@sn - номер закла
@engname - английское имя закла
@rusname - русское имя закла
@argument - все, что идет после названия закла
@victim - на кого кастится закл
@vobj - предмет, на который кастится закл (для некоторых заклов, которые хотя
и кастятся на предмет, этоа переменная не устанавливается по тех.
причинам).
@dirname - название направления, в котором кастится закл
@revdirname - название направления, обратного dirname.
@toroom - внум комнаты, в направлении которой было колдунство.

	delay
	не работает
	
	

	drop
	Срабатывает, когда этот объект бросают.
	Число – вероятность.
	@victim – кто бросил;

@vobj – что бросил.

	eat
	Срабатывает при съедании обжекта.
	Число – вероятность.
	@victim – кто съел;

@vobj – что съел.

	enter
	Срабатывает при входе чара в руму, в т.ч. и на невидимых вроде.
	Число – вероятность.
	@victim – кто вошел.

	get
	Срабатывает, когда этот объект берут.
	Число – вероятность.
	@victim – кто бросил;

@vobj – что бросил.

	give
	Срабатывает, когда этот объект дают.
	Число – вероятность.
	@victim – кто дал;

@vobj – что дал.

	look
	Срабатывает, когда на обжект смотрят.
	Число – вероятность.
	@victim – кто смотрит;

@vobj – на что смотрят.

	open
	Срабатывает на открывание обжекта.
	Число – вероятность.
	@victim – кто открыл;

@vobj – что открыл.

	push

pull
	Срабатывают на толкание и дергание рычага соответственно.
	Число – вероятность.
	@victim – кто толкал;

@vobj – что толкал.

	put
	Срабатывает, когда обжект во что-нибудь кладут. При этом обжект не кладется, надо пофорсить в скрипте.
	Число – вероятность.
	@victim – кто кладет;

@vobj – что кладет.

	put_in
	Срабатывает, когда в обжект что-нибудь кладут. При этом обжект не кладется, надо пофорсить в скрипте.
	Число – вероятность.
	@victim – кто кладет;

@vobj – что кладет.

	random
	Срабатывает раз в тик.
	Число – вероятность.
	Нет.

	sit

rest

sleep

stand
	Срабатывает при попытке (даже неудачной) соответствующего действия с объектом.
	Число – вероятность.
	@victim – кто;

@vobj – куда.

	speech
	Срабатывает, когда чар произносит фразу.
	Строка – сказанная фраза.
	@victim – кто сказал;

@argument – что сказал.

	turn
	Срабатывает на команду «повернуть».
	Число – вероятность.
	@victim – кто повернул;

@vobj – что повернул.

	wear
	Срабатывает, когда объект одевают.
	Число – вероятность.
	@victim – кто одел;

@vobj – что одел.


8. Разбор полетов
В этой главе будут разобраны простые и не очень простые скрипты, встречавшиеся в реальной жизни, типичные задачи и т.п. Сначала попытайтесь решить задачу сами. Если не получится с первого раза – попытайтесь восстановить действия, которые должен совершить компилятор. Если не найдете баг – сверьте с примером.
8.1 Простейшие скрипты

Приветствие с выводами параметров чара
Задача: чар входит в комнату, при этом стоящий в комнате моб говорит ему его имя, имя его божества и его профессию.

Давайте сначала разберемся с выбором триггера. Очевидно, что нужен тот, который реагирует на попадание в комнату. В моб-триггерах у нас есть аж четыре таких триггера – greet, enter, grall и enterall. Какой из них выбрать? Прежде всего, чтобы поприветствовать чара, моб должен его видеть (по-моему, логично). Поэтому отметаем варианты grall и enterall. Если нам нужно какое-то определенное направление, при входе с которого приветствует моб, тогда выбираем enter. Если нам нужно, чтобы он реагировал всегда – greet. Теперь сам скрипт:

ска Привет, @victim@!

ска Твое божество - @victim->deity@!

ска Твоя профессия - @victim->class@!

Вот и все. Пояснять тут особо нечего. Если что-то непонятно, смотри главу про доступ к переменным.

Теперь давайте немного усложним задачу. В ответе будет скрипт, взятый из Инсельгарда.

Задача: поприветствовать чара, сказать его профессию, если профессия воинская – сказать, что он может идти дальше, если магическая – сказать, что дальше идти нельзя.

Триггер такой же, как и в предыдущем примере. Однако сам скрипт немного осложнится условиями:
if (!is_npc(@victim)) then

  ска Приветствую тебя, @victim@!

  улыб 

  ска Я вижу, что ты @victim->rusclass@.

  if ((@victim->class == 'thief') || (@victim->class == 'warrior') || (@victim->class == 'ranger') || (@victim->class == 'paladin') || (@victim->class == 'nazgul') || (@victim->class == 'monk')) then
    ска Можешь пройти в тренировочный класс на севере, там тебя могут чему-нибудь научить.

  else

    ска Лучше тебе покинуть это место. Здесь должны находиться лишь бойцы, не маги.

  endif

endif
Обратите внимание – в начале скрипта стоит проверка is_npc. Скрипт продолжит работу, только если вошедший – чар. Вторая проверка построена на использовании оператора «ИЛИ». То есть блок кода внутри ифа выполнится, если хотя бы одно из условий верно.

Пропуск чара в комнату при выходе. Запрет входа.

Теперь перейдем к следующему скрипту, который стоит рядом с этим в Инсельгарде. В воинскую гильдию пускают только воинские профессии. Мы уже поприветствовали персонажа, теперь наша задача – не допустить в гильдию магов, но пропустить воинов. Итак,

Задача: персонаж пытается идти на север. Если у него воинская профессия – пропустить его. Если нет – не пропускать.

Сначала перед нами стоит выбор типа скрипта и триггера. Тут можно пойти несколькими путями. Можно запретить ходьбу с помощью моб-триггера exall. Можно также запретить просто команду в комнате с помощью рум-триггера command. Я, когда строил Инсельгард, пошел вторым путем. О нем и расскажу. Соответственно, на комнату ставим триггер command с аргументом north. Он не даст персонажу выполнить эту команду. Итак, скрипт:

if ((@victim->class == 'thief') || (@victim->class == 'warrior') || (@victim->class == 'ranger') || (@victim->class == 'paladin') || (@victim->class == 'nazgul') || (@victim->class == 'monk')) then
  room force секретарь ска Пожалуйста, проходи, там тебя встретит мастер гильдии.

  room echoat @victim@ Секретарь провожает тебя в тренировочный зал.

  room echoaround @victim@ Секретарь провожает @victim->nvin@ в тренировочный зал.

  room transfer @victim@ 9152

  if (is_affected(@victim, 'flying')) then

    room at 9152 echoaround @victim@ @victim@ прилетает с юга.

  else

    room at 9152 echoaround @victim@ @victim@ приходит с юга.

  endif

else

  room force секретарь ска Куда?!

  room force секретарь ска Ну-ка вон из гильдии бойцов!

  room echoaround @victim@ Секретарь выпроваживает @victim->nvin@.

  room echoat @victim@ Секретарь выпроваживает тебя.

  room transfer @victim@ 9058

endif
В первой строке скрипта стоит уже знакомая нам проверка. Далее комната общается от имени секретаря – сначала форсит его сказать, потому выдает эхи. Разницы никакой. Потом чара трансферит в нужную комнату. Обратите внимание, что трансферит только чара. Трансфер очарок и т.п. был бы чревать абьюзами, так как заодно могло бы переместить и персонажей с другими профессиями. В комнате, куда затрансферили чара, выдается сообщение (с учетом эхи, которая должна была быть). Обратите внимание, что просто пофорсить чара идти на север нельзя – тоже чревато теми же абьюзами. Дальше, вроде, все понятно.

Убийство персонажа

Частенько в скриптах бывает необходимо убить персонажа. Однако дело это не такое простое, как кажется на первый взгляд.

Задача: убить персонажа при произнесении им фразы 123. Сделать это двумя способами.
Как я говорил уже, не все так просто. Есть некоторые не очевидные сразу тонкости. Во-первых, есть две функции, с помощью которых можно убить персонажа: room damage и damage(@ch, @dam). Какую из них выбрать? Раньше второй не было. Функция room damage учитывает все резисты, заклинания на чаре и т.п. Реальных повреждений чар получает куда меньше. Функция же damage(@ch, @dam) наносит повреждения без учета. Однако давайте попробуем убить чара обоими методами.

Триггер очевиден – speech с аргументом 123. Пусть это будем рум-скрипт. Начнем с простого варианта. Сколько надо снять жизни с чара, чтобы он умер? Очевидно, что жизни надо загнать в минус. Так что снимаем все его хиты + еще сколько-то. Еще сколько-то должно зайти за порог оглушения чара. Дело в том, что когда у чара от нуля до где-то -10 хп, то он лежит оглушенный, и его можно спасти. Так что, нам надо перешагнуть через этот порог. Скрипт простой и состоит из одной строчки:

damage(@victim, @victim->max_hit + 500).

Это гарантировано убьет чара. Теперь давайте перейдем к более сложному варианту – так приходилось убивать чаров раньше. Как известно уже, мы не можем точно рассчитать, сколько урона принесет room damage. Поэтому мы будем дамажить чара, пока он не умрет. А слово пока переводится на английский как while. Намек понят? Используем цикл while. Однако, какой взять критерий, что чар уже мертв? Ведь хитов меньше нуля у него никогда не будет, а если продамажить уже умершего чара, то он умрет еще раз! Ответ прост. Надежным параметром, который всегда меняется при смерти чара, можно считать номер комнаты, где он находится. Очевидно, что этот номер комнаты не будет совпадать с номером комнаты, из которой вызывается скрипт. А пока он не умер, он будет стоять в комнате, где скрипт. Все просто. Вот скрипт:

while (@victim->in_room->vnum == @room->vnum) do
room damage #victim# 10000 10000 lethal
endwhile
Обратите внимание на два момента – в функции дамаги виктим взят в решетки, чтобы случайно не продамажить другого (ведь в комнате могут быть мобы с одинаковыми именами), и в конце функции дамаги стоит слово lethal – иначе чар не умрет. Вот и весь скрипт.
Давайте немного усложним задачу. Сделаем руму тикерной ловушкой – на тике она будет убивать всех чаров в руме.

Убивать будем первым способом – без заморочек. Очевидно, что сменится условие, когда скрипт запускается, то есть триггер. Раз в тик у нас работает триггер random. Однако, в нем нет встроенной переменной персонажа. Придется  ее выцеплять самому. Что же мы будем делать? Переменная персонажа в комнате - @room->people. У нас появится переменная с типом mob_data, из которой мы сможем выцепить следующего персонажа, и так, пока они не кончатся в комнате (заметьте, я опять употребил слово «пока»(). Однако, и тут не без сложностей – дело в том, что когда мы убьем персонажа, он уже будет в другой комнате, и наш скрипт может начать убивать ни в чем не повинных граждан. Эту проблему можно решить, если мы введем промежуточную переменную. Вот скрипт:

var @victim.

var @oldvictim.

@victim := @room->people.

while (!is_nullvar(@victim)) do

@oldvictim := @victim.
@victim := @victim->next_in_room.

damage(@oldvictim, @oldvictim->max_hit + 500).

endwhile
Думаю, этот скрипт я уже объяснил достаточно.

Колдовство на персонажа
Очень часто бывает нужно наложить на виктима какой-либо эффект. Это у нас делается с помощью заклинаний(
Задача: чар смотрит на объект. Чар при этом слепнет.

Очевидно, это это обж-скрипт, и триггер – look. Подковырка в том, что с первого раза эффект вряд ли наложится. Поэтому придется его пытаться наложить, пока (() персонаж не ослепнет. Очевидно, кстати, что слепой чар на обжект не посмотрит. Вот скрипт:

while (!is_affected(@victim, ‘blind’)) do

obj cast слепота #victim# 60

endwhile
if (is_affected(@victim, ‘blind’)) then

obj echoat #victim# Яркий свет фонаря ослепляет тебя!

obj echoaround #victim# Яркий свет фонаря ослепляет @victim->nvin@!

endif
С while все понятно. Однако обратите внимание на if, который идет после него. Ведь мы можем за все итерации цикла так и не наложить слепоту. Поэтому обязательно следует проверить, слеп ли чар, прежде чем выдавать ему эху.

На этом, пожалуй, я простые скрипты закончу. Мусолить их можно бесконечно – на каждую команду по парочке. Однако, все они однотипны. Поэтому я начинаю следующую часть. Не приступайте к ее чтению, пока не сможете писать простенькие скрипты с условиями и короткими циклами.

8.2 Сложные скрипты. Вызовы скриптов.

Хочется обратить внимание, что тут ни один скрипт и ни одна задача не вымышлена. Такие скрипты встречаются и не редко. В хорошей зоне без них не обойтись. И вовсе не всегда они висят на оверных квестовых мобах (хотя, конечно, чаще всего). Все скрипты сперты из мира. Да простят меня их владельцы за копирайт, хотя я постараюсь разбирать преимущественно мои. Абьюзеры, радуйтесь, короче( Вот скрипт на вход в зону клана «Клинки Хлама». Сначала попробуйте разобраться в нем сами.
1. if (is_npc(@victim) || order(@mob) != 0) then
 2.   exit 

 3. endif

 4. @x := @mob->in_room->people.

 5. while (!is_nullvar(@x)) do

 6.   @y := @x->next_in_room.

 7.   if (!is_npc(@x) & !(@x->clan == 'Клинки Хлама' || @x->leader->clan == 'Клинки Хлама' || is_immortal(@x)) || is_killer(@

x)) then

 8.     if (is_killer(@x)) then

 9.       say Убийц велено не пускать. Воля Бессмертных.

10.     else

11.       ска @x->nime@, немедленно покинь это место!

12.     endif

13.     if (is_sitting(@x) || is_sleeping(@x)) then

14.       mob force #x# встать
15.     endif

16.     mob echoat #x# @mob->nime@ выталкивает тебя с территории клана.

17.     mob echoaround #x# @mob->nime@ выталкивает @x->nvin@ с территории клана.

18.     mob at 2411 mob echo @x@ прибегает с юга.

19.     mob mtransfer #x# 2411

20.   endif

21.   @x := @y.

22. endwhile

23. if (!(@victim->clan == 'Клинки Хлама' || @victim->leader->clan == 'Клинки Хлама' || is_immortal(@victim))) then

24.   exit 

25. endif

26. if (@victim->in_room->vnum == @mob->in_room->vnum) then

27.   mob oload 2400

28.   отпереть юг

29.   открыть юг

30.   if (@victim->clan == 'Клинки Хлама') then

31.     if (@victim->sex == 1) then

32.       ска Проходи, брат!

33.     else

34.       ска Проходи, сестра!

35.     endif

36.   elseif (is_immortal(@victim)) then

37.     ска Мы всегда вам рады!

38.   elseif (@victim->leader->clan == 'Клинки Хлама') then

39.     ска Проходи, но соблюдай порядок!

40.   endif

41.   mob force #victim# south

42.   закрыть юг
43.   запереть юг
44.   mob remove #mob# 2400

45. endif
Попробовали разобраться? Теперь я объясню моменты, в которых вы можете и не разобраться. Строчки 1-3 – затычка, на кого скрипт работать не должен. Не срабатывает он на мобов, а также он не работает два раза – там стоят два моба, и он будет работать только на одном из них. Сначала скрипт проходится по всем, кто в комнате (похожая ситуация у нас была в простом скрипте с тикерной ловушкой). А потом совершает действия непосредственно с виктимом.

Вот другая система скриптов – система пропусков для клана «Сумеречный Легион». Скрипт на вход в клан там похож на предыдущий. Однако там можно проходить тому, у кого есть пропуск, выписанный членом клана. У Клинков же зато можно проводить своих одногруппников. Будете делать кланзону – решайте сами, что удобней. Может, придумаете и что-нибудь свое. Вот скрипт на выписывание пропуска:

1. if ((@victim->clan == 'Сумеречный Легион') || (is_immortal(@victim))) then

 2.   if (@argument == '') then

 3.     room force стражник сказать Кому пропуск-то выписывать?

 4.     exit 

 5.   endif

 6.   var @target.

 7.   @target := get_char_world(@argument, 0).

 8.   if (get_char_world(@argument, 0) == NULL) then

 9.     room force стражник ска Слушай, не морочь мне голову, таких тут нет!

10.   else

11.     room force #victim# ска Выпиши пропуск @target->ndat@.

12.     if (@target->sex == 1) then

13.       room force стражник ска Сейчас выпишу ему пропуск.

14.     else

15.       room force стражник ска Сейчас выпишу ей пропуск.

16.     endif

17.     room oload 2729 '@victim@'

18.     var @obj.

19.     @obj := @victim->carrying.

20.     if (@obj->vnum == 2729) then

21.       if (!is_nullvar(@obj->owner)) then

22.         set_owner(@obj, @target->name).

23.         set_timer(@obj, 24 - hour).

24.       endif

25.     endif

26.     room echo Стражник что-то пишет на свитке и запечатывает его сургучом.

27.     room echoaround #victim# Стражник дает #victim->ndat# пропуск.

28.     room echoat #victim# Стражник дает тебе пропуск.

29.     room force стражник ска Пропуск действителен до полуночи.

30.   endif

31. else

32.   room force стражник ска А ты что за цаца, пропуски выписывать?

33.   room force стражник хмык

34. endif
Пропуска могут выписывать только члены клана и имморталы. Скрипт стоит на триггере new_cmd. Соответственно мы можем работать с аргументом команды. Аргументом и будет служить имя чара, которому надо выписать пропуск. Функция get_char_world ищет чара в мире. Если он есть – ему выписывается пропуск. Дальше все довольно банально, кроме строчек 17-25. На этих строчках грузится пропуск, и ему присваивается владелец и таймер. После того, как таймер дойдет до нуля, обжект рассыпется. А владельца надо приписывать, чтобы чары не проходили с чужими пропусками – проверка пропусков идет по внуму и владельцу. Выцепляем мы объект в 19-й строчке – он становится первым, который несет наш виктим. Объект может не загрузиться чару в инвентарь. Поэтому сначала идет проверка, тому ли объекту мы присваиваем владельца. Потом, если это вдруг окажется другой уже выписанный пропуск, тогда нам тоже не надо менять ни таймер, ни владельца. Вот, пожалуй и все, что требует отдельного объяснения. Для полноты картины все-таки приведу скрипт для проверки пропусков:
1. if (!is_killer(@victim)) then

 2.   if ((@victim->clan == 'Сумеречный Легион') || (@victim->level >= 52) || (wears(@victim, 2729))) then

 3.     if (@victim->level >= 52) then

 4.       сказ Всегда вам рады!

 5.     else

 6.       if (@victim->clan == 'Сумеречный Легион') then

 7.         if (@victim->sex == 1) then

 8.           ска Проходи, брат!

 9.         else

10.           ска Проходи, сестра!

11.         endif

12.       endif

13.     endif

14.     кланят 

15.   else

16.     ска Куда?! Эт территория клана!

17.     mob mtransfer '@victim@' 2730

18.   endif

19. else

20.   ска Убийц пускать не велено!

21.   mob mtransfer '@victim@' 2730

22. endif

Это скрипт на greet у моба стоит. Затычки на порядок моба не стоит( Поэтому скрипт спамный. Если чару в зону нельзя – он просто выталкивается. Кстати, триггер greet и подобные ему не работают, когда чар сбегает из соседней клетки. Поэтому были абьюзы. Для этого нужен и скрипт, пускающий внутрь:

1. if (!is_killer(@victim)) then

 2.   if ((@victim->clan == 'Сумеречный Легион') || (@victim->level >= 52)) then

 3.     mob echoat '@victim@' Стражник пропускает тебя внутрь.

 4.     mob echoaround '@victim@' Стражник пропускает @victim->nvin@ внутрь.

 5.     отк юг
 6.     mob force '@victim@' south

 7.   else

 8.     var @obj.

 9.     @obj := get_equip(@victim, 'hold').

10.     if (is_nullvar(@obj)) then

11.       ска У тебя нет пропуска! Внутрь не пущу!

12.       exit 

13.     endif

14.     if ((@obj->vnum == 2729)) then
15.       mob echoaround #victim# Стражник смотрит на пропуск @victim->nrod@.

16.       mob echoat #victim# Стражник смотрит на твой пропуск.

17.       if (@obj->owner == @victim->name) then

18.         mob echoat '@victim@' Стражник пропускает тебя внутрь.

19.         mob echoaround '@victim@' Стражник пропускает @victim->nvin@ внутрь.

20.         отк юг
21.         mob force '@victim@' south

22.       else

23.         ска Это не твой пропуск! Ворюга!

24.       endif

25.     else

26.       ска У тебя нет пропуска! Внутрь не пущу!

27.     endif

28.   endif

29. else

30.   ска Убийц пускать не велено!

31. endif

Скрипт снова проверяет наличие пропуска. И тут уже проверяется владелец пропуска. Это как раз для абьюзеров.

Позволю себе маленькое лирическое отступление. Учитесь читать чужие скрипты! Это очень полезно, прежде всего, чтобы самому научиться их читать. Так что, не ленитесь и разберите все эти три скрипта от начала до конца по руководству. Может, найдете багу – тогда скажите о ней( Я вот только что одну нашел. В эхе стояла переменная в решетках. Должно было быть имя, а переменная в решетках – это число. Поэтому и в эху выдало бы число.

Следующий скрипт стоит на моем троне. Как известно, сидеть на нем кому попало нельзя. А если сядет кто-то не тот – улетит куда подальше. Скрипт несколько устарел – сейчас он делается куда проще.

1. var @rrr.

 2. @rrr := 1.

 3. var @roo.

 4. var @tmp.

 5. if ((@victim->name == 'Элдерос')||(@victim->name == 'Эйлинель')) then

 6.   obj echo Трон начинает скрипеть, и, кажется, сейчас развалится.

 7.   obj echoat @victim@ Тебя окутывает мягкое синее сияние.

 8. obj echoaround @victim@ @victim->nvin@ окутывает мягкое синее сияние.

 9.   delay(2).

10.   obj echo Сверху прорезается небольшой лучик света, постепенно освещая всю комнату.

11. else

12.   obj echo Трон окружается багровым сиянием.

13.   obj echo Вся комната на мгновение погружается в темноту.

14.   delay(2).

15.   obj echoaround '@victim@' Когда мрак рассеивается, ты видишь, что @victim->nime@ исчез.

16.   obj echoat '@victim@' Когда мрак рассеивается, ты обнаруживаешь себя в другом месте.

17.   obj echoaround '@victim@' Ты слышишь приглушенный крик @victim->nrod@.

18.   wait_state(@victim, 20).

19.   obj echoat '@victim@' Твое тело сводит страшная судорога!

20.   while (@rrr != 0) do

21.     delay(0).

22.     @tmp := random(25000).

23.     @roo := is_room(@tmp).

24.     if (@roo != 0) then

25.       @rrr := 0.

26.       obj transfer '@victim@' @tmp@

27.       break

28.     endif

29.   endwhile

30. endif

Сначало все просто (посмотрите на 18-ю строчку, кстати, чтобы поучиться пользоваться wait_state). Сложный скрипт начинается c 20-й строки. Там выбирается рандомная комната в мире. Сейчас для этого уже есть специальная функция, тогда же, когда я писал скрипт, ее не было, поэтому я сделал, как смог. В теории, если ну ОЧЕНЬ не повезет, может рухнуть мир. Но это вряд ли. Функция is_room возвращает переменную room_data. И пока у нас переменная с комнатой пуста, выбираются рандомные внумы. Можете взять себе задачу – переписать скрипт так, чтобы он соответствовал всем новым требованиям (тут надо сразу несколько моментов исправить). Также в скрипте есть бага. Можно много раз садится на трон, и скрипт будет каждый раз запускаться. Подумайте, как ее исправить.
Следующий скрипт – скрипт на обновленной секире «Ярость Кхвайма». Не очень, конечно, хотелось его выкладывать, но…

1. delay(2).

 2. @victim := @obj->wear_by.

 3. if (@victim) then

 4.   if ((@victim->deity == 'Кхвайм') || is_npc(@victim)) then

 5.     @target := @victim->fighting.

 6.     if (@target) then

 7.       if (!can_see(@victim, @target)) then

 8.         exit 

 9.       endif

10.       @damage := (@victim->str - @target->str) * (get_skill(@victim, 'топоры') / 100).

11.       if (@damage <= 0) then

12.         exit 

13.       endif

14.       @miss := (get_skill(@target, 'увертка') - get_skill(@victim, 'топоры')) + (@victim->dex - @target->dex).

15.       if (is_npc(@target)) then

16.         @damage *= 15.

17.       else

18.         @damage *= @target->max_hit / 100.

19.       endif

20.       obj echoat #victim# Тебе чудится, что на секире на мгновение показалось лицо Кхвайма.

21.       if (@miss > 0 && random(100) < @miss) then

22.         obj echoat #victim# Ты пытаешься нанести сильный удар по @target->ndat@, но промахиваешься.

23.         obj echoat #target# @victim@ пытаешься нанести по тебе сильный удар, но промахивается.

24.         obj echonear #victim# #target# @victim@ пытается нанести по @target->ndat@ сильный удар, но промахивается.

25.         exit 

26.       endif

27.       if (@victim->sex == 1) then

28.         obj echoat #target# В глазах @victim->nrod@ загорается божественный огонь, и он наносит тебе сильный удар секирой

.

29.         obj echoat #victim# Воодушевленный этим, ты наносишь по @target->ndat@ сильнейший удар.

30.       else

31.         obj echoat #target# В глазах @victim->nrod@ загорается божественный огонь, и она наносит тебе сильный удар секиро

й.

32.         obj echoat #victim# Воодушевленная этим, ты наносишь по @target->ndat@ сильнейший удар.

33.       endif

34.       obj echonear #victim# #target# @victim@ наносит сильнейший удар секирой по @target->ndat@.

35.       damage(@target, @damage).

36.     endif

37.   endif

38. endif

Скрипт – небольшая модель умешки. Висит на обжекте на триггере рандом, то есть срабатывает раз в тик. Обратите внимание на третью строку. Там в условии указана только переменная @victim. В данном случае, но не всегда, это равносильно выражению is_nullvar(@victim). Ведь если переменная будет пустой, то значит и выражение будет ложью( В остальном скрипт довольно простой.

Вот затычка недоработки с морским драконом. Довольно сложный скрипт с использованием массивов. Основной сложностью было то, что из двух скриптов, установленных на триггер random, работает только первый. Поэтому пришлось объединять два скрипта в один. Задача его – во-первых, топить чаров без масок, во-вторых – топить очарок, владелец которых где-то далеко (так как сбежать им некуда, и способ добывания кольца становился очень простым).

1. @tmp := @room->people.

 2. if (!@tmp) then

 3.   exit 

 4. endif

 5. if (!wears(@tmp, 8400) && !is_npc(@tmp) && !is_immortal(@tmp)) then

 6.   room echoat @tmp@ Вода потоком устремляется в твои легкие!

 7.   room echoat #tmp# Ты захлебываешься и умираешь!

 8.   room echoaround #tmp# @tmp@ захлебывается в воде и мертвым опускается на дно.

 9.   damage(@tmp, @tmp->max_hit).

10.   room damage #tmp# 40 50 lethal

11. endif

12. if (!is_npc(@tmp)) then

13.   @chars[@tmp->nime] := 1.

14. endif

15. while (@tmp := @tmp->next_in_room) do

16.   if (!wears(@tmp, 8400) && !is_npc(@tmp) && !is_immortal(@tmp)) then

17.     room echoat @tmp@ Вода потоком устремляется в твои легкие!

18.     room echoat #tmp# Ты захлебываешься и умираешь!

19.     room echoaround #tmp# @tmp@ захлебывается в воде и мертвым опускается на дно.

20.     damage(@tmp, @tmp->max_hit).

21.     room damage #tmp# 40 50 lethal

22.   endif

23.   if (!is_npc(@tmp)) then

24.     @chars[@tmp->nime] := 1.

25.   endif

26. endwhile

27. @tmp := @room->people.

28. while (@tmp) do

29.   if (is_npc(@tmp) && !@chars[@tmp->leader->nime] && @tmp->vnum != 15401) then

30.     room echoaround #tmp# @tmp->nvin@ затягивает в бездонную глубину озера.

31.     @oldtmp := @tmp.

32.     @tmp := @tmp->next_in_room.

33.     room purge #oldtmp#

34.   else

35.     @tmp := @tmp->next_in_room.

36.   endif

37. endwhile
На самом деле, ничего особо сложного в нем тоже нет. Чтобы узнать, у каких очарок хозяин стоит в руме, а у каких – нет, использован массив. Индексами массива становятся имена чаров, стоящих в руме, а значениями элементов – единица. Потом если имени лидера в массиве нет, то есть в массиве не существует переменной с индексом имени лидера, то очарка пуржится.

